


## 9 . 光学で使用する特殊関数

```
[> restart;
```

Maple には光学に必要な特殊関数がすでに組み込まれており、それを数値的にも数式的にも扱うことができます。

### rect 関数

```
[> rect:=t->Heaviside(t+a)-Heaviside(t-a);  
      rect:=t→Heaviside(t+a) − Heaviside(t−a) (1.1)  
> plot(eval(rect(t),a=1/2),t=-1..1,title="rect function");
```



```
[> with(inttrans);
```

```
> fourier(rect(t),t,w);  

$$\frac{2 \sin(a w)}{w} \quad (1.2)$$

```


結果は sinc 関数と呼ばれる関数になります。

```
> sinc:=t->sin(t)/t;  

$$\text{sinc} := t \rightarrow \frac{\sin(t)}{t} \quad (1.3)$$

```

```
> plot(sinc(t),t=-15..15,title="sinc function");
```



## ▼ デルタ関数

```
> Dirac(t);
```

$\text{Dirac}(t)$

(2.1)

[Diracはその定義 (t=0でのみ値を持つ) よりプロットでは描けません。

```
> fourier((2.1),t,w);
```

1

(2.2)

## ▼ ガウス分布(確認)

レーザー光など、瞳上の強度分布が正規分布を持つ場合などに出てきます。

```
> restart;
```

```
> with(Statistics):
```

[ガウス分布を生成します。

```
> Gau:=RandomVariable(NormalDistribution(mu,sigma));
```

$Gau := \_R$

(3.1)

[確率密度関数の計算


```
> PDF(Gau,u);
```

$$\frac{1}{\sqrt{2\pi}\sigma} e^{-\frac{1}{2}\frac{(u-\mu)^2}{\sigma^2}} \quad (3.2)$$

```
> eval((3.2),{u=1,sigma=1});
```

$$\frac{1}{\sqrt{2\pi}} e^{-\frac{1}{2}(1-\mu)^2} \quad (3.3)$$

```
> plot((3.3),mu=-5..5,title="Gaussian distribution");
```



3次元


```
> Gau2:=(x,y)->A * exp(-(x^2+y^2)/w);
```

$$Gau2 := (x, y) \rightarrow A e^{-\frac{x^2+y^2}{w}} \quad (3.4)$$

```
> eval(Gau2(x,y),{A=1,w=1});
```

$$e^{-x^2-y^2} \quad (3.5)$$

```
> plot3d((3.5),x=-2..2,y=-2..2,axes=boxed);
```



## ▼ Bessel 関数

一次のベッセル関数

```
> J1:=x->BesselJ(1,x);
```


$$J1 := x \rightarrow \text{BesselJ}(1, x) \quad (4.1)$$

例えば、瞳が円形、理想レンズ（波面収差ゼロ）の回折像は、Bessel 関数を使った次のような式で与られます。

```
> U:=2*J1(r)/r;
```

$$U := \frac{2 \text{BesselJ}(1, r)}{r} \quad (4.2)$$


```
> plot(U,r=-15..15,title="diffraction image");
```



3次元

```
> plot3d(eval(U,r=sqrt(x^2+y^2)),x=-10..10,y=-10..10,grid=[30,30],  
title="3D diffraction image",axes=boxed);
```

3D diffraction image



## ▼ Fresnel 関数

フレネルコサイン関数

```
> FresnelC(t);
```

$FresnelC(t)$  (5.1)


フレネルサイン関数

```
> FresnelS(t);
```

$FresnelS(t)$  (5.2)

コルニユの渦巻き

```
> plot([(5.1), (5.2), t=-5..5], title="Cornu Spiral");
```



*Copyright © CYBERNET SYSTEMS CO., LTD. 2009 All rights reserved.*